HEWLETT PACKARD

Application Note 377-1
Automatic Frequency Profiling Of Chirped Radar Pulses

Using the HP 5361A 20 GHz Pulse/CW Microwave Counter



Single-Shot Radar Profiling

If your application is a single-shot pulse or requires profiling resolution of less than 20 ns, the HP 5361A is not the instrument of choice for you. Instead, choose one of HP's Frequency and Time Interval Analyzers such as the HP 5371A.

The HP 5371A, along with the HP 5364A Microwave Mixer/Detector, provides powerful measurement capabilities, comprehensive phase and timing analysis, and built-in graphics. Contact your local HP representative or see Product Note 5371A-1 for more details.

Introduction

■ The HP 5361A 20 GHz Pulse/CW Microwave Counter is well suited to fully characterizing frequency and timing parameters of pulsed microwave signals. Whatever your application — from testing frequency-agile radio signals or pulse compression techniques to analyzing frequency variations in Doppler radar waveforms — the time-dependent frequency changes within the signal can be easily measured and recorded.

The HP 5361A can be used independently or teamed with a high performance computer such as the HP 9000 Series 200/300. This Application Note describes the use of the HP 5361A and HP 9000 Computer, in conjunction with an HP 5359A Time Synthesizer and an HP digitizing oscilloscope, to obtain the frequency profile of an unknown signal. The method presented in this Note describes instrument control via the HP Interface Bus (HP-IB), thus yielding automatic, reliable, repeatable, and precise data collection.

Included in this Note is a description of the pulse frequency profiling technique and a step-by-step guide to automatic profiling of your signal. An abbreviated version of the listing of the HP BASIC program which controls the process via the HP 9000 Computer appears at the end of this Note to facilitate system integration and reduce the complicated measurement series to an automated, user-friendly program. The complete program (including simple graphics) can be ordered free with the enclosed card.

The HP 5361A: a Useful Tool for Pulsed Radar Profiling

■ The HP 5361A is the first fully automatic pulse microwave counter to offer high-performance CW measurements. When used on a pulsed signal, it allows accurate determination of pulse width and average frequency, as well as pulse repetition interval (PRI), pulse repetition frequency (PRF), and pulse off time. In addition, the counter can be used for frequency profiling, or quantifying the frequency variations within a signal.

The HP 5361A, used with the profiling method described in this Note, brings the following benefits to radar chirp analysis:

- **Fully automated data-taking and analysis**
- **Ease of use**
- **True, programmable measurement resolution**
- **High measurement throughput**
- **Easy to interpret data output**
- **A cost effective measurement solution**

Characterizing a Pulsed Signal

■ A typical pulsed radar signal is shown in Figure 1. The HP 5361A automatically measures:

- **average signal frequency within the pulses**
- **pulse width (PW)**
- **pulse repetition interval (PRI)**
- **pulse repetition frequency (PRF)**
- **pulse off time**

In addition, the frequency variations within the pulse can be determined with the use of an external gating pulse.



FIGURE 1. A Typical Pulsed Radar Signal

A time synthesizer, synchronized with the counter and thus the input signal, produces a measurement gate of selectable width and delay. When applied to the external gate of the counter, it can be made to step through the signal burst to obtain the frequency within each gate (see Figure 2). The summation of this data provides an accurate frequency profile of the incoming signal.



FIGURE 2. Pulse Profiling Using an External Gate

Required Equipment

■ The automatic measurement system for performing frequency profiling, as shown in Figure 3, includes the following instruments:

HP 5361A 20 GHz Pulse/CW Microwave Counter.

HP 9000 Series 200 or 300 Technical Computer (or an IBM® PC/AT ™ - Compatible Computer and HP 82300B HP BASIC Language Processor). This high-performance workstation acts as the instrument controller for programmable frequency profiling. (NOTE: This computer facilitates the frequency profiling measurements described in this Application Note, but is not required; manual profiling can be accomplished by systematically increasing the external gate delay and noting the measured frequency at each delay. See the section "Going Further" for more details.)

HP 5359A Time Synthesizer.* This highly accurate frequency and pulse generating system is used to produce the time-delayed external measurement gate for the HP 5361A.



FIGURE 3. Measurement Setup Block Diagram

** While this Application Note is written specifically for a test setup using the HP 5359A Time Synthesizer to generate the measurement gate, any delaying pulse generator can be used. With some minor modification to the attached BASIC Program, any of the following generators can be substituted for the HP 5359A: HP 8112A, HP 8115A, HP 8131A, HP 8160A, and HP 8161A. The tradeoff is timing jitter and expense.*

HP Digitizing Oscilloscope (optional). This oscilloscope is used to monitor the gating and measurement process through the HP 5361A SCOPE-VIEW output, thus graphically indicating the location of an internal or external gate within the pulse burst.

Accessories. In addition, two HP-IB cables, one microwave cable appropriate to the signal frequency, and three BNC cables for the gating process are required.

Measurement Setup

■ Connect the test equipment as shown in Figure 3. The following is a step-by step guide to connecting and running the system to perform pulse frequency profiling.

1. INPUT FREQUENCY: While the HP 5361A counter provides two input ports for measuring various frequencies, only Input 1 is utilized for pulsed signals. Connect the test signal to Input 1.

2. EXTERNAL GATE SYNC: Connect the rear panel PULSE OUT of the HP 5361A to the EXT TRIGGER of the HP 5359A in order to synchronize the external gate with the test signal.

3. EXTERNAL GATE POLARITY: Set the polarity of the HP 5359A pulse output to COMP, POS, and the HP 5359A EXT TRIGGER slope to positive.

4. EXTERNAL GATE: Connect the 50 OHM OUTPUT of the HP 5359A (providing the delayed gate) to the GATE/ARM IN of the HP 5361A.

5. HP-IB: Connect the HP-IB cable between the HP 9000 Computer and the HP 5361A counter, and between the counter and the HP 5359A Time Synthesizer. Set the HP-IB address on the HP 5361A to "14", and on the HP 5359A to "9".

6. (OPTIONAL) In order to view the gating/measurement process within the pulse burst, connect the HP 5361A's SCOPE-VIEW to the input of the oscilloscope. See the HP 5361A 20 GHz Pulse/CW Microwave Counter Demo/Getting Started Guide (literature number, 5952-7985) for a recommended setup.

Check the following Table to ensure that the test signal and pulse generator output fall within the operating range of the HP 5361A.

INPUT	SPECIFICATION
TEST SIGNAL	
Input Frequency	500 MHz to 20 GHz
Input Power	-25 dBm to + 7 dBm (-20 dBm to +7 dBm for frequencies > 12.4 GHz)
Maximum FM Modulation	10 MHz *
Maximum Pulse Repetition Frequency	2 MHz
Minimum Pulse Width	100 nsec
HP 5359A TIME SYNTHESIZER	
Output Level	0 to 2.5 V
Output Polarity	Positive, comp
Minimum Ext. Gate Width	20 nsec (manual mode)

* *The maximum FM modulation can be increased to 50 MHz if the counter is used in MANUAL acquisition mode. To do this, make the following changes in the provided BASIC program:*

1. *In "INITIALIZE INSTRUMENTS" section:*
   - *Delete the lines OUTPUT @ Counter; "SET"*
     OUTPUT @ Counter; "RESET"
   - *Change the line : OUTPUT @ Counter; "AUTO"*
     To: OUTPUT @ Counter; "MANUAL"
2. *Manually input the frequency (within 1 MHz) at the HP 5361A front panel.*

TABLE 1. Operating Specifications

To perform the actual measurements using the automated program, load and run the available software. If you desire to manually profile your pulse, refer to the "Going Further" section at the end of this Note.

Program Description

■ The HP BASIC program listed at the end of this Application Note provides an interactive method to fully characterize an unknown signal. It performs all measurements and outputs the results in both tabular and graphic form (if requested).*

After initializing the test equipment over the HP-IB, the program does the following:

(1) Acquires the signal in AUTO acquisition mode
(2) Measures and records average frequency, PRF, PRI, pulse width, and pulse off time
(3) Programs the counter to MANUAL acquisition mode with center frequency equal to average frequency
(4) Requests the desired gate width and number of steps to be taken within the pulse for frequency characterization
(5) Places the initial gate at the start of the signal pulse and begins measurement
(6) Measures enough signal pulses to obtain the specified frequency resolution, then records the measurement
(7) Continues to step the measurement gate through the input signal, recording each measured frequency
(8) Displays the results
(9) Optionally graphs the results (frequency within the measurement gate vs. step number); (graphics routines are not listed in this Note, but are provided in the available software).

A word should be said regarding the described method of generating the external gate. The HP 5361A PULSE OUT signal is a negative-true pulse corresponding to the pulse envelope of the input signal. The positive-going edge of this envelope (correlating to the end of the signal pulse) is used to trigger the HP 5359A Time Synthesizer. The Synthesizer then generates a pulse that is delayed from this point by a time equal to the pulse off time plus an additional variable delay. It is this final pulse which then becomes the external gate for the HP 5361A (see Figure 2). By creating the gate in this manner, it becomes possible to measure the signal frequency at the start of the pulse.

In order to accurately measure an externally gated signal, the HP 5361A requires that the end of the gate occur prior to the end of the signal. Due to this requirement, a signal containing excessive amounts of jitter or instability may result in erroneous measurements. If this appears to be a problem in your application, use the SCOPE-VIEW feature to verify that the gate ends before the end of the signal pulse.

**It should be noted that this program has specific application to signals with a constant Pulse Repetition Interval (PRI) and pulse width. If the signal contains jitter or stagger in the pulse location, the program and external gate generator must be changed to compensate for this. See "Going Further" on pg. 8 for more detail.*

Monitoring the Gating Process with an Oscilloscope (Optional)

■ As mentioned earlier, the SCOPE-VIEW port of the HP 5361A can be connected to an oscilloscope to view the measurement gating process. When using external gating, this unique capability can be a very useful tool to graphically show the location of the measurement gate within the signal pulse.

The enclosed HP BASIC program may step the measurement gate through the pulse too quickly to allow any meaningful data to be seen on the oscilloscope. If it is desired to visually monitor this gating process, the gates may either be manually stepped through the signal, or the enclosed program may be modified by changing the 'WAIT 1' commands to 'WAIT 10' or longer. In addition, if the resolution is increased, the instrument will require longer measurement times, and the signal will remain on the oscilloscope for a longer time.

In this mode, the oscilloscope will show the pulsed signal with the gated portion offset from the ungated portion (See Figure 4). This allows visual inspection of the gating function location, thus providing confidence in the measurements.



FIGURE 4. SCOPE-VIEW Output

Going Further

■ You may want to alter the pulse profiling method described here to focus in greater detail on concerns specific to your application. The following chart summarizes some of the possible modifications and their consequences.

IF YOU WANT	THEN. . .	BUT. . .
Higher frequency resolution	Increase the resolution in the HP BASIC program or on the face of the HP 5361A (for manual profiling applications)	The measurement time will increase in proportion to the square of the desired increase in frequency resolution
A more accurate representation of the instantaneous frequency within the pulse	Decrease the width of the measurement gate, down to the minimum of 20 ns, and increase the number of measurement steps (optional)	The measurement time may increase if more steps are requested
Faster measurement time (or on the face of the	Decrease the resolution in the HP BASIC program HP 5361A for manual applications) OR	The frequency resolution will degrade
	Respond to the "Number of Measurement Steps" prompt with a smaller number	The frequency measurements within the pulse will occur at larger time intervals
Frequency linearity statistics	Perform a linear best-fit analysis on the measured frequency-vs-time data	Increased programming time
To profile radar pulses that have varying PRIs	Remove initial programmable delay (See Fig. 2) so that the pulse generator starts with a delay of 0 seconds. Also, change the EXT TRIGGER slope of the pulse generator to "NEGATIVE"	The leading edge of the pulse will not be profiled
To manually profile a pulse without the use of the available software	Make the following settings on the HP 5359A: •set EXT TRIGGER slope to negative •set the output delay to zero •set pulse width to the desired gate width Set the HP 5361A to external gating; manually step the ext gate through your pulse by increasing the output delay on the HP 5359A.	Extensive user interaction is required, and profiling will take more time than with the automated software

HP Basic Automated Frequency Profiling Program*

```
10  !
20  !      AUTOMATED PULSE FREQUENCY PROFILING
30  !    USING THE HP 5361A 20 GHz PULSE/CW
40  !          MICROWAVE COUNTER
50        Automatically measures average frequency within
70  !   the pulse, PRF, PRI, Pulse width, and Pulse off-time.
80  !
90  !            PROFILING MODE:
100 !     Requests the width of the measurement gate which
110 !     is to be stepped through the signal, along with
120 !     the number of steps to take.  Outputs all data to
130 !     the user via the computer screen.
150 !
170 !
180 !              INITIALIZE INSTRUMENTS
200 ASSIGN @Counter TO 714                    ! * HPIB:  Counter = 14
210 ASSIGN @Synth TO 709                      ! * Time Synthesizer = 9
220 REMOTE @Counter
230 REMOTE @Synth
240 !
250 DIM Freq(2000)
260 DIM Time(2000)
270 Fmin=1.0E+25                              ! * Initialize min. frequency
280 Fmax=1.0E-25                              ! * ... and max. frequency
290 OUTPUT @Counter;"SET"
300 OUTPUT @Counter;"RESET"                   ! * Reset counter; set initial params:
310 OUTPUT @Counter;"FMRATE,NORMAL"           ! *  Normal FM rate
320 OUTPUT @Counter;"SAMPLE,FAST"             ! *  Fast sample rate
330 OUTPUT @Counter;"RESOL,4"                 ! *  10 kHz resolution
340 OUTPUT @Counter;"AUTO"                    ! *  Automatic counter acquisition
350 OUTPUT @Synth;"W0E-06"                    ! * Reset msmt. gate delay to zero
360 !
380 !           MEASURE PULSE SIGNAL PARAMETERS
400 CLEAR SCREEN
430 PRINT "    AUTOMATED PROFILING OF A CHIRPED RADAR SIGNAL *"
450 OUTPUT @Counter; "Chirp On"
460 OUTPUT @Counter;"FREQ"                    ! * Measure average frequency
470 WAIT 1
480 REPEAT
490   ENTER @Counter;Ave_freq
500 UNTIL (Ave_freq>0) AND (Ave_freq<1.0E+38)
520 OUTPUT @Counter;"PWID"                    ! * Measure Pulse width
530 WAIT 1
540 REPEAT
550   ENTER @Counter;Pulse_width
560 UNTIL (Pulse_width>0) AND (Pulse_width<1.0E+38)
580 OUTPUT @Counter;"OFFT"                    ! * Measure Pulse off time
590 WAIT 1
600 REPEAT
610   ENTER @Counter;Off_time
620 UNTIL (Off_time>0) AND (Off_time<1.0E+38)
640 OUTPUT @Counter;"PRI"                     ! * Measure Pulse Repetition Interval
650 WAIT 1
660 REPEAT
670   ENTER @Counter;Pri
680 UNTIL (Pri>0) AND (Pri<1.0E+38)
700 OUTPUT @Counter;"PRF"                     ! * Measure Pulse Repetition Frequency
710 WAIT 1
720 REPEAT
730   ENTER @Counter;Prf
740 UNTIL (Prf>0) AND (Prf<1.0E+38)
760 PRINT "    CHARACTERISTICS OF PULSED WAVEFORM"
```

**This software is offered at no charge as an example of the techniques described in this Application Note. Software performance is not warranted by Hewlett-Packard.*

```
770 PRINT ""
780 PRINT "    AVERAGE FREQUENCY (Hz) : ";Ave_freq
790 PRINT "    PRF (Hz)               : ";Prf
800 PRINT "    PRI (sec)              : ";Pri
810 PRINT "    OFF TIME (sec)         : ";Off_time
820 PRINT "    PULSE WIDTH (sec)      : ";Pulse_width
830 PRINT ""
850 !
860 ! PERFORM PROFILING MEASUREMENTS
870 !
890 PRINT "            PULSE PROFILE MEASUREMENTS"
900 PRINT " The entire pulse will now be profiled; external gate steps will"
910 PRINT " be spaced evenly through pulse to provide frequency measurements"
930 PRINT "     Enter the desired gate width in microseconds"
940 PRINT "     (must be less than the pulse width of ";Pulse_width*1.E+6;"µsec):"
950 INPUT Width
960  Nsteps=INT(Pulse_width/(Width*1.E-6))
970 PRINT "       ... and the desired number of steps for frequency"
980 PRINT "           determination (default = ";Nsteps;"):"
990 INPUT Nsteps
1010 PRINT "YOU HAVE CHOSEN ";Nsteps;" STEPS OF ";Width;" MICROSECONDS EACH."
1050 !
1060 !   DATA ACQUISITION
1070 !
1080 OUTPUT @Counter;"MANUAL";Ave_freq
1090 OUTPUT @Counter;"GATE,EXT"          ! * Set counter for external trigger
1100 OUTPUT @Synth;"W",Width;"E-06"      ! * Output gate width to synthesizer
1110  P=MIN(Pulse_width, PRI-Offtime)
1120  Dly2=((.98*P)-(width*1E-6))/(Nsteps-1)
1130 FOR I=1 TO Nsteps                   ! * Step msmt gate through pulse:
1140   Dly=(Off_time+(I-1)*Dly2)*1.E+6   !   Set delay
1150   Time(I)=Dly-(Off_time*1.E+6)
1160   OUTPUT @Synth;"D",Dly;"E-06"      !   ... and output to synthesizer
1170   OUTPUT @Counter;"FREQ"            ! * Measure frequency within gate
1180   WAIT 1
1190   REPEAT
1200     ENTER @Counter;Freq(I)
1210   UNTIL (Freq(I)>0) AND (Freq(I)<1.0E+38)
1220   IF Freq(I)<Fmin THEN Fmin=Freq(I)  ! * Record minimum frequency
1230   IF Freq(I)>Fmax THEN Fmax=Freq(I)  !   ...  maximum frequency
1240   PRINT I,Freq(I)
1260 NEXT I
1290 Timemin=Time(1)
1300 Timemax=Time(Nsteps)
1380 !  PRINT OUT TABLE OF RESULTS
1440 PRINT "  SUMMARY OF PULSE FREQUENCY PROFILING "
1460  PRINT "_____________________________________________________________________"
1470 PRINT "TIME AFTER START   FREQUENCY     TIME AFTER START   FREQUENCY"
1480 PRINT " OF PULSE (usec)     (MHz)        OF PULSE (usec)     (MHz)"
1490  PRINT "_____________________________________________________________________"
1500 Format:   IMAGE 2(XX,DDDD.DD,XXXXXXXXX,DDDDD.DDDD,XXXXXXXX)
1510 FOR J=1 TO INT(Nsteps+1/2) STEP 2
1520   IF (J+1)>Nsteps THEN Freq(J+1)=0.
1530   PRINT USING Format;Time(J),1.E-6*Freq(J),Time(J+1),1.E-6*Freq(J+1)
1540 NEXT J
2300 !
2310 Finish:   OUTPUT @Counter;"GATE,INT"
2320         OUTPUT @Counter;"AUTO"
2350 PRINT "  END OF RADAR PULSE PROFILING PROGRAM "
2360 PRINT "              (To run again, press RUN)"
2370 LOCAL 714                           ! * Return instruments to
2380 LOCAL 709                           !   local control
2390 END
```

For more information, contact the factory-authorized distributor or HP Sales Office listed here.

United States:
Hewlett-Packard Company
Laser Marketing, MS55/16
5301 Stevens Creek Blvd.
Santa Clara, CA 95052
Tel. 408-246-4300

Canada:
Hewlett-Packard Ltd.
6877 Goreway Drive
Mississauga, Ontario
L4V1M8
Tel. 416-678-9430

United Kingdom:
Hahn & Kolb Limited
Leicester Road
Rugby, Warwickshire CV21 1NY
Tel. 0788-77288

West Germany:
Hahn & Kolb
Borsigstrasse 50
7000 Stuttgart 30
Tel. 0711-2004-0

France:
S.I.M.C.I.S.A.
Tour Alpha
128, Av Des Champs Lasniers
91940 Les Ulis
Tel. 446-18-75

Australia:
Hibbert Machine Tool Service Co. Limited
145 Glen Eira Road
East St. Kilda 3183
Tel. 03-528-3550

South Africa:
Traconsa (Pty) Limited
P.O. Box 1472
Kempton Park 1620
Tel. 011-975-7040

Europe:
Hewlett-Packard S.A.
7, rue du Bois-du-Lan
CH-1217 Meyrin 2
Geneva, Switzerland

Japan:
Yokogawa-Hewlett-Packard Limited
29-21, Takaido-Higashi 3-Chome
Suginami-ku, Tokyo 168
Tel. (03) 331-6111

Far East:
Hewlett-Packard Asia Headquarters
47th F, China Resources Bldg.
26 Harbour Road, Wanchai
Hong Kong

Other International Areas:
Hewlett-Packard Company
Intercontinental Headquarters
3495 Deer Creek Rd.
Palo Alto, CA 94304
U.S.A.

HEWLETT PACKARD

Printed in U.S.A.
Data Subject to Change
5952-7987
June 1989